# How to get 'exact' expression for result

I have a result from Mathematica for which want to have the exact expression, i.e. all of the arguments of type # replaced. It is given by:

{ a > 0,
Inequality[ 0, Less, r, LessEqual, Root[a^3 #1^2 + a^2 #1^4 - 22 a #1^6 + #1^8 & , 5]]}


where a and r are the only two variables. What exactly does the #1 mean here? To which variables does it refer? How do I make Mathematica replace all these place holders with its corresponding variable?

-
Use e.g. Root[a^3 #1^2 + a^2 #1^4 - 22 a #1^6 + #1^8 &, 5] // ToRadicals. However in general one cannot express Root objects in terms of radicals since it is mathematically impossible. See e.g. How do I work with Root objects? –  Artes Jul 12 '13 at 11:20

Root objects are just exact expressions, if possible one can express them in terms of radicals (e.g. with ToRadicals). To deal with inequalities the best approach might be Reduce:

ToRadicals @ Reduce[{ a > 0,
Inequality[0, Less, r, LessEqual,
Root[a^3 #1^2 + a^2 #1^4 - 22 a #1^6 + #1^8 &, 5]]},
a] // ComplexExpand

 r > 0 && a >= -(r^2/3) + 2/3 Sqrt[67] r^2 Sin[1/3 ArcTan[227/(3 Sqrt[127947])]]

% // TraditionalForm


and we can get an arbitrarily precise numerical approximation in a standard way, e.g. 40-digit precission:

%% // N[#, 40] &

r > 0 && a >= 0.04555316444285376876486224542203853971094 r^2

-
Thank you! One more question, for which variable does #1 in this case stand? –  bonanza Jul 12 '13 at 11:26
It is an abstract variable see e.g. Slot. Root[a^3 #1^2 + a^2 #1^4 - 22 a #1^6 + #1^8 &, 5] denotes 5-th root of the polynomial a^3 x^2 + a^2 x^4 - 22 a x^6 + x^8, you can get it with e.g. a^3 #1^2 + a^2 #1^4 - 22 a #1^6 + #1^8 &[x] –  Artes Jul 12 '13 at 11:28